

## Hovey 達の式を用いた塩化カリウム水溶液の飽和水蒸気圧を求める 計算プログラム

300°Cを超える高温条件での塩化カリウム水溶液の飽和水蒸気圧に関するHovey達の式を本サイト(<http://www.hyogo-u.ac.jp/sci/yshibue/solution.html>)内の文書で解説した。飽和水蒸気圧の計算プログラムをテキスト形式で保存して本サイト内で示している。このプログラムは澁江(2012, p. 85–91)が示したものに数箇所の修正を加えたものである。

計算プログラムをテキスト形式で保存して本サイト内で示している。URLは次の通りである。<http://www.hyogo-u.ac.jp/sci/yshibue/solution.html>。ここでは、計算プログラムへの入力と出力、プログラム中の変数について解説する。そして、最後にプログラムのリストを示す。文書の1行文字数の関係で、プログラムの1つのlineが文書では複数行にまたがっていることがある。プログラムでは行番号が先頭に付くので、行番号ごとにひとまとまりのlineになっていると理解してほしい。

### 1. 入力と出力

この文書ではプログラムへの入力例と出力例を示す。そして、プログラム中で用いている変数を説明する。プログラムを起動すると、以下の問いが出てくる。問いを(1), (2), (3), (4)で示し、入力方法をそれぞれについて説明する。

(1) “Input temperature (°C)?”

摂氏温度で温度を入力する。入力した温度が300°C未満だと、プログラムは温度の再入力を求めるようになっている。

(2) “Input molality of KCl in liquid phase?”

液相中での塩化カリウムの質量モル濃度を入力する。

(3) “Do you input initial estimates for DRL, DRV, and YV? Input Y (or y) or N (or n)?”

この問いは、液相中での水の密度と関連する値である $r^{\text{liquid}}$ 値(DRL), 気相中での水の密度と関連する値である $r^{\text{vapor}}$ 値(DRV), 気相中での塩化ナトリウムの $y$ 値(YV)を全て入力するかどうかに関するものである。これら3つの未知数についての初期推定値を手入力する時はYあるいはyを入力する。これ以外の場合はNあるいはnを入力する。

手入力すると答えた場合には、まず、“DRL?”と問いかけがあるので $r^{\text{liquid}}$ 値を入力する。次に、“DRV?”と問いかけがあるので $r^{\text{vapor}}$ 値を入力する。次に、“YV?”と問いかけがあるので $y^{\text{vapor}}$ 値を入力する。Nあるいはnと入力すると、初期推定値をプログラムが計算する。

(4) 全ての計算が終了すると、新たに“Input temperature (°C)?”と問いが出てくる。続けて計算する時は(1)から(3)の要領で入力していく。計算を終了させる時は、“Ctrl”と“c”を同時に押す。

温度を300°C, 濃度を1 mol kg<sup>-1</sup>, 初期推定値の手入力を行わないとした時の入力と出力をそれぞれ表1と表2に示す。

表1 温度を300°C, 濃度を1 mol kg<sup>-1</sup>, 初期推定値の手入力を行わないとした時の入力画面

---

```
run
Input temperature (deg C)? 300
Input molality of KCl in liquid phase? 1
Do you input initial estimates for DRL, DRV, and YV? Input Y (or y) or N (or n)? n
Input temperature (deg C)?
```

---

表2 温度を300°C, 濃度を1 mol kg<sup>-1</sup>, 初期推定値の手入力を行わないとした時の出力画面

---

```
Temperature(° C)=300.000      molality of NaCl= +1.000000

Eps=+1.00D-005
DRLMAX=+3.50 DRLMIN=+1.50 DRVMAX=+1.00 DRVMIN=+0.02 YVMAX=+1.70D-002 YVMIN=+1.00D-012
Initial estimate
BC(1)=+2.3308D+000 BC(2)=+1.4339D-001 BC(3)=+3.0291D-008
Initial FSUM=+4.728114D-004

ICON= 3 FSUM=+4.847746D-005 ITERATION= 1
BC(1)=+2.3297D+000 BC(2)=+1.3674D-001 BC(3)=+2.0880D-008

ICON= 3 FSUM=+6.225447D-008 ITERATION= 2
BC(1)=+2.3297D+000 BC(2)=+1.3707D-001 BC(3)=+2.2305D-008

ICON= 1 FSUM=+7.942032D-014 ITERATION= 3
BC(1)=+2.3297D+000 BC(2)=+1.3707D-001 BC(3)=+2.2355D-008

ICON= 0 FSUM=+2.183228D-023 ITERATION= 4
BC(1)=+2.3297D+000 BC(2)=+1.3707D-001 BC(3)=+2.2355D-008

Solution
BC(1)= +2.3296830857D+000
BC(2)= +1.3707026654D-001
BC(3)= +2.2354732014D-008

Pressure liquid=+8.354D+001 vapor=+8.354D+001
Density liquid=+8.061D-001 vapor=+4.414D-002
```

---

表2で示した出力について説明を加える。

- (1) 1行目の温度と濃度は入力値を示す。
- (2) 3行目は, 逐次近似計算の収束条件 (Epsの値) を示す。値は10のべき乗を用いる指数形式で表している。1.00D-005と記しているのでEpsは $10^{-5}$ である。
- (3) 4行目は, 逐次近似計算の際のDRLとDRVとYVの最大値と最小値を示している。DRLの最大値をDRLMAX, DRLの最小値をDRLMIN, DRVの最大値をDRVMAX, DRVの最小値をDRVMIN, YVの最大値をYVMAX, YVの最小値をYVMINとして表している。YVMAXの値は,  $y^{\text{liquid}} - 0.001$ に取り, YVMINの値を $10^{-12}$ に取っている。YVMAXの値を指定しないと, 収束値が $y^{\text{liquid}}$ と等しくなる場合が出てくるのでこのようにした。その他の値は温度によって変えている。350°C以下では, DRLMAX=3.5, DRLMIN=1.5, DRVMAX=1, DRVMIN=0.02にしている。350°Cより高温では, DRLMIN=1.0, DRVMAX=1.5にしている。
- (5) 5行目と6行目は, DRL, DRV, YVに関する初期推定値を示している。BC(1)はDRL, BC(2)はDRV, BC(3)はYVの初期推定値である。
- (6) 7行目は, 初期推定値を用いて求めた次の計算値(FSUM)を示している。

$$FSUM = \left( \frac{p^{\text{vapor}}}{p^{\text{liquid}}} - 1 \right)^2 + \left( \frac{\mu_w^{\text{vapor}}}{\mu_w^{\text{liquid}}} - 1 \right)^2 + \left( \frac{\mu_s^{\text{vapor}}}{\mu_s^{\text{liquid}}} - 1 \right)^2$$

- (7) 9行目と10行目は, 初期推定値から求められる新しい推定値とFSUMの値を示している。ICONとして示した値は収束条件を満たしていない未知数の数を示し, ITERATIONとして示した数は逐次近似計

算の回数を表す。12行目から19行目についても同様である。

(8) 21行目から24行目は、DRL, DRV, YVの計算結果を示している。BC(1)は $r^{\text{liquid}}$ , BC(2)は $r^{\text{vapor}}$ , BC(3)は $y^{\text{vapor}}$ の計算結果である。

(9) 26行目は、DRL, DRV, YVの計算結果を用いて計算した圧力(単位はbar)を液相と気相に分けて示している。

(10) 27行目は、DRLとDRVの計算結果を用いて計算した液相と気相の密度(単位は $\text{g cm}^{-3}$ )を液相と気相に分けて示している。なお、逐次近似の回数が100回を超えてしまった場合には、BC(1), BC(2), BC(3)の計算結果に「Function improvement possible」と印字するようになっている。

入力段階で過飽和状態になる濃度を入力した場合には、DRLとDRVとYVに関する問いかけの前に計算が始まる。例えば、300°Cで塩化カリウムの質量モル濃度を20と入力した場合、計算が始まって表3のように出力が出る。

表3 温度を300°C, 濃度を20 mol kg<sup>-1</sup>とした時の出力画面

---

Temperature(° C)=350.000 molality of KCl=+20.000000

Supersaturated (or saturated) with KCl

Pressure = +4.888D+001 Solubility=+15.63

---

表3で示した出力について説明を加える。

(1) 1行目の温度と濃度は入力値を示す。

(2) 3行目は、塩化カリウムに対して過飽和状態あるいは飽和状態であることを表している。

(3) 4行目は、Hovey et al. (1990)から計算した飽和水蒸気圧が48.88 barであることと、この温度における飽和濃度が15.63 mol kg<sup>-1</sup>であることを表している。Hovey達の式を用いて密度を計算することが可能ではあるが、結果の正確さが不明であるのでここでは求めていない。

## 2. プログラム使用上の留意点

この計算プログラムではHaar et al. (1984)の状態方程式を用いて計算した圧力の単位をbarに取っており、MPaからの換算のために変数FPを用いている。10 cm<sup>3</sup> bar = 1 Jの関係を用いて、体積と圧力の積をエネルギーに換算するために、line 10450で変数VC(純水の臨界モル体積)の値を10で割っている。

気液平衡計算を行う時に、圧力や液相中と気相中の水と塩化ナトリウムの化学ポテンシャルの計算値を液相と気相の密度や気相に含まれている塩化ナトリウムの物質質量(モル)で偏微分する計算が必要となる。本計算プログラムでは、これらすべての偏微分係数の計算値を差分で求めている。

気相における塩化カリウムの $y$ 値の上限(YVMAX)を $YVMAX = YL - 0.001$ で与えている(リスト中のline 22600)。したがって、気液二相平衡状態の計算も $y^{\text{liquid}} - y^{\text{vapor}} \geq 0.001$ の領域に限定されている。つまり、臨界点の計算や臨界点付近での計算を行うことができない。line 22600で示した計算式を変えることで臨界点付近における気液二相平衡状態の計算が可能になる。ただし、YVMAXがYLと等しいとおくと、臨界点ではないのに気相の組成が液相の組成と等しくなって計算が終了することがある。本計算プログラムで使用する気体定数の値は、Haar et al. (1984)が用いた値とは異なっている。そこで、質量モル濃度として0を入力した時(あるいは誤って負の値を入力した時)には質量モル濃度を再入力するようにしている。

解説で記したように、液相の密度の初期推定値を求める計算式は350°C程度までの実験値から求めている。したがって、初期推定値を求める計算式を350°Cを超える温度で使用すると、初期推定値の不確かさが大きくなり、計算に失敗することがある。このような場合には、濃度はそのままにして温度を350°Cあるいはそれ以下に指定して、いったん低温での飽和水蒸気圧を計算する。求めることができたDRL, DRV, YVの値を用いて、温度だけを少し高くして計算する。新たに求めることができたDRL, DRV, YVの値を用いてさらに温度を高くする。以上の操作を繰り返して、求めようとし

ている温度での飽和水蒸気圧を計算する。臨界点付近では、1°C刻みで温度を上げていくようなことをしないと、計算が収束しない。

### 3. プログラム中の変数

プログラムで用いている変数（文字変数とFORループの制御変数を除く）の意味を表にして以下に示す。変数の中にはプログラム中で二通りの意味で用いられているものがある。そこで、このような変数については、変数名の後の括弧内に行番号を示している。なお、配列変数の添字に大文字のIを用いているが、その添字は係数に付ている下付き文字*i*と同義である。また、HGK式で用いられている変数やHGK式を用いて計算している変数は「変数の意味」欄に「(HGK式)」と記している。

プログラム中の変数	変数の意味
AA	1 (HGK式)
AAD(I)	$\alpha_i$ (HGK式)
AAT(I)	$\beta_i$ (HGK式)
AB	$A_{\text{base}}/RT$ (HGK式)
AD	$A/RT$ (HGK式)
ADZ(I)	$\rho_i$ (HGK式)
AI	$A_{\text{ideal gas}}/RT$ (HGK式)
AL	解の計算に取り入れる縮小因子
AR	$A_{\text{residual}}/RT$ (HGK式)
ATT	$\beta_i$ (HGK式)
ATZ(I)	$T_i$ (HGK式)
B1	$b$ (HGK式)
B10	Hovey et al. (1990)の状態方程式中の係数
B11	Hovey et al. (1990)の状態方程式中の係数
B2	$\bar{B}$ (HGK式)
B20	Hovey et al. (1990)の状態方程式中の係数
BA1,BA2,BA3,BA4, BA5	塩化カリウム水溶液の密度に関するHaas式で使用する係数
BASEF	$p_{\text{base}}/\rho RT$ (HGK式)
BBC(1)	$r^{\text{liquid}}$ の計算値
BBC(2)	$r^{\text{vapor}}$ の計算値
BBC(3)	$y^{\text{vapor}}$ の計算値
BC(1)	$r^{\text{liquid}}$ の計算値
BC(2)	$r^{\text{vapor}}$ の計算値
BC(3)	$y^{\text{vapor}}$ の計算値
BP(1)	$b_0$ (HGK式)
BP(2)	$b_1$ (HGK式)
BP(3)	$b_2$ (HGK式)
BP(4)	0 (HGK式)
BP(5)	$b_3$ (HGK式)

---

BP(6)	$b_4$ (HGK式)
BP(7)	$b_5$ (HGK式)
BP(8)	0 (HGK式)
BP(9)	0 (HGK式)
BP(10)	0 (HGK式)
BQ(1)	$B_0$ (HGK式)
BQ(2)	0 (HGK式)
BQ(3)	$B_1$ (HGK式)
BQ(4)	$B_2$ (HGK式)
BQ(5)	$B_3$ (HGK式)
BQ(6)	$B_4$ (HGK式)
BQ(7)	$B_5$ (HGK式)
BQ(8)	0 (HGK式)
BQ(9)	0 (HGK式)
BQ(10)	0 (HGK式)
C(I)	$C_i$ (HGK式)
COEF(I,J)	最小二乗法による計算に用いる行列
D	計算過程で用いる密度の値 (HGK式)
D1	Haas式から求められる水1 kgを含む水溶液の体積(cm <sup>3</sup> )
DAQ	Haas式から求められる塩化カリウム水溶液の密度(g cm <sup>-3</sup> )
DD	密度の計算値 (HGK式)
DDZ	$\rho_i$ (HGK式)
DEL	$\delta_i$ (HGK式)
DELDL	数値微分するための変数(10 <sup>-6</sup> )
DELDV	数値微分するための変数(10 <sup>-6</sup> )
DEX	$\delta_i^j \exp(-\alpha_i \delta_i^{k_i})$ (HGK式)
DIFMYUS	$\mu_s^{\text{vapor}} - \mu_s^{\text{liquid}}$
DIFMYUWATER	$\mu_w^{\text{vapor}} - \mu_w^{\text{liquid}}$
DIFP	$p^{\text{vapor}} - p^{\text{liquid}}$
DIFPLV	$p^{\text{liquid}} - p^{\text{vapor}}$
DL	計算過程で用いる液相の密度の値
DMYUSDDL	$\left( \frac{\partial \mu_s^{\text{liquid}}}{\partial r^{\text{liquid}}} \right)_{T, r^{\text{vapor}}, y^{\text{vapor}}}$
DMYUSDDV	$\left( \frac{\partial \mu_s^{\text{vapor}}}{\partial r^{\text{vapor}}} \right)_{T, r^{\text{liquid}}, y^{\text{vapor}}}$

---

---

DMYUSDY	$\left(\frac{\partial \mu_s^{\text{vapor}}}{\partial y^{\text{vapor}}}\right)_{T, r^{\text{liquid}}, r^{\text{vapor}}}$
DMYUWATERDDL	$\left(\frac{\partial \mu_w^{\text{liquid}}}{\partial r^{\text{liquid}}}\right)_{T, r^{\text{vapor}}, y^{\text{vapor}}}$
DMYUWATERDDV	$\left(\frac{\partial \mu_w^{\text{vapor}}}{\partial r^{\text{vapor}}}\right)_{T, r^{\text{liquid}}, y^{\text{vapor}}}$
DMYUWATERDY	$\left(\frac{\partial \mu_w^{\text{vapor}}}{\partial y^{\text{vapor}}}\right)_{T, r^{\text{liquid}}, r^{\text{vapor}}}$
DPDDL	$\left(\frac{\partial p^{\text{liquid}}}{\partial \rho^{\text{liquid}}}\right)_{T, y^{\text{liquid}}}$
DPDDV	$\left(\frac{\partial p^{\text{vapor}}}{\partial \rho^{\text{vapor}}}\right)_{T, y^{\text{vapor}}}$
DPDY	$\left(\frac{\partial p^{\text{vapor}}}{\partial y^{\text{vapor}}}\right)_{T, r^{\text{vapor}}}$
DRL	液相の $r$ 値
DRL0	Haas式から求められる液相の $r$ 値
DRLDEC	DRLを求めるための変数
DRLMAX	DRLの最大値
DRLMIN	DRLの最小値
DRV	気相の $r$ 値
DRV0	Wagner and Pruss (1993)の式で近似した気相の $r$ 値
DRVMAX	DRVの最大値
DRVMIN	DRVの最小値
DV	計算過程で用いる気相の密度の値
DWATERV	Wagner and Pruss (1993)の式で計算した純水だけからなる気相の密度
E	$e^{-p}$ (HGK式)
EPS	逐次近似計算の収束条件
EX1	$-\alpha_i \delta_i^{k_i}$ (HGK式)
EX2	$-\beta_i \tau_i^2$ (HGK式)
FCT	$\frac{\delta_i^l}{\rho_i} (l_i \delta_i^{-1} - \alpha_i k_i \delta_i^{k_i - 1}) \rho^2 \exp(-\alpha_i \delta_i^{k_i} - \beta_i \tau_i^2)$ (HGK式)
FD	1.0 (HGK式)
FH	18.01534
FP	圧力の単位をbarにするための換算定数(10.0) (HGK式)
FSUM	$\left(\frac{p^{\text{vapor}}}{p^{\text{liquid}}} - 1\right)^2 + \left(\frac{\mu_w^{\text{vapor}}}{\mu_w^{\text{liquid}}} - 1\right)^2 + \left(\frac{\mu_s^{\text{vapor}}}{\mu_s^{\text{liquid}}} - 1\right)^2$

---

---

FSUMI	逐次近似計算の過程における $\left(\frac{p^{\text{vapor}}}{p^{\text{liquid}}}-1\right)^2 + \left(\frac{\mu_w^{\text{vapor}}}{\mu_w^{\text{liquid}}}-1\right)^2 + \left(\frac{\mu_s^{\text{vapor}}}{\mu_s^{\text{liquid}}}-1\right)^2$
G1	11 (HGK式)
G2	133/3 (HGK式)
GASCON	気体定数を水のモル質量で割った値( $0.461522 \text{ J g}^{-1} \text{ K}^{-1}$ )。
GD	$G/RT$ (HGK式)
GF	7/2 (HGK式)
GI	$A_{\text{ideal gas}}$ を計算するための式で用いる変数 (HGK式)
GIB	純水のギブスエネルギー
GIBL	純水が気液二相平衡条件下にある時の液相のギブスエネルギー
GIBLD	密度がRHOLDの時の純水のギブスエネルギー
GIBV	純水が気液二相平衡条件下にある時の気相のギブスエネルギー
GIBVD	密度がRHOVDの時の純水のギブスエネルギー
H2O	水のモル質量(= $18.01534 \text{ g mol}^{-1}$ )
HGKG(I)	$g_i$ (HGK式)
ICON	収束条件を満たしていない未知数の数
II(I)	$k_i - 1$ ( $i = 1, \dots, 36$ )あるいは $k_i$ ( $i = 37, \dots, 40$ ) (HGK式)
IMAX	逐次近似計算の上限回数
INC	36 (HGK式)
ITER	逐次近似計算の回数
JJ(I)	$l_i + 1$ ( $i = 1, \dots, 36$ )あるいは $l_i$ ( $i = 37, \dots, 40$ ) (HGK式)
K	$k_i$ (HGK式)
KCL	塩化カリウムのモル質量(= $74.551 \text{ g mol}^{-1}$ )
KK	未知数の数
KM	$l_i$ (HGK式)
KP1	未知数の数に1を加えた値
L	$l_i + 1$ (HGK式)
M	質量モル濃度
MYULS	液相中での塩化カリウムの化学ポテンシャルの値から $g^*$ を引いたもの
MYUVS	気相中での塩化カリウムの化学ポテンシャルの値から $g^*$ を引いたもの
MYULWATER	液相中での水の化学ポテンシャル( $\text{J mol}^{-1}$ )
MYUVWATER	気相中での水の化学ポテンシャル( $\text{J mol}^{-1}$ )
PL	液相の組成と密度の推定値から計算できる飽和水蒸気圧
PLD	$\left(\frac{\partial p^{\text{liquid}}}{\partial r^{\text{liquid}}}\right)_{T, y^{\text{liquid}}}$ を求めるための変数

---



---

PVLS	気相と液相と固相が平衡状態にある時の飽和水蒸気圧
PRES	$p$ (HGK式)
PV	気相の組成と密度の推定値から計算できる飽和水蒸気圧
PVD	$\left(\frac{\partial p^{\text{vapor}}}{\partial r^{\text{vapor}}}\right)_{T,y^{\text{vapor}}}$ を求めるための変数
Q	$p_{\text{residual}}$ (HGK式)
Q5	$\left(\frac{\partial p_{\text{residual}}}{\partial \rho}\right)_T$ (HGK式) であるが, ここでは計算していない。
Q10 (line 16800)	$\rho^2 e^{-\rho}$ (HGK式)
Q10 (line 18500)	$\delta_i^{l_i} \exp(-\alpha_i \delta_i^{k_i} - \beta_i \tau_i^2)$ (HGK式)
Q20	$1 - e^{-\rho}$ (HGK式)
QK	$k_i$ (HGK式)
QKM	$l_i$ (HGK式)
QM	$l_i \delta_i^{-1} - \alpha_i k_i \delta_i^{k_i-1}$ (HGK式)
QP(line 17500)	$g_i \rho^2 e^{-\rho} (1 - e^{-\rho})^{k_i-1} \left(\frac{T_0}{T}\right)^{l_i}$ (HGK式)
QP(line 18650)	$\sum_{i=37}^{40} \left[ \frac{g_i}{\rho_i} (l_i \delta_i^{-1} - \alpha_i k_i \delta_i^{k_i-1}) \delta_i^{l_i} \rho^2 \exp(-\alpha_i \delta_i^{k_i} - \beta_i \tau_i^2) \right]$ (HGK式)
QR(I)	$\rho^2 e^{-\rho} (1 - e^{-\rho})^{i-2}$ (HGK式)
QT(I)	$\left(\frac{T_0}{T}\right)^{i-2}$ (HGK式)
QZR(I)	$\rho^2 e^{-\rho} (1 - e^{-\rho})^i$ (HGK式)
QZT(I)	$\left(\frac{T_0}{T}\right)^{i-1}$ (HGK式)
R	気体定数(= 8.3144 J K <sup>-1</sup> mol <sup>-1</sup> )
RHOC	臨界点における純水の密度(= 0.322 g cm <sup>-3</sup> )
RHOL	液相の密度の計算値(g cm <sup>-3</sup> )
RHOLD	$\left(\frac{\partial p^{\text{liquid}}}{\partial r^{\text{liquid}}}\right)_{T,y^{\text{liquid}}}$ を求めるための変数
RHOLIQUID	液相の密度(g cm <sup>-3</sup> )
RHOV	気相の密度の計算値(g cm <sup>-3</sup> )
RHOVAPOR	気相の密度(g cm <sup>-3</sup> )

---

---

RHOVD	$\left(\frac{\partial p^{\text{vapor}}}{\partial r^{\text{vapor}}}\right)_{T,y^{\text{vapor}}}$ を求めるための変数
RT	$RT$ (J mol <sup>-1</sup> )
RTG	$RT$ (J g <sup>-1</sup> ) (HGK式)
S	気相と液相と固相が平衡状態にある時の液相中の塩化ナトリウムの濃度 (wt.%)
SREF	HGK式では基準状態を三重点での液相にしている。この時にエントロピーの計算値が0になるようにしている。Haar et al. (1984)はサブルーチン THERMDTで無次元化したエントロピー(SD)の計算値からSREFの値を引いている。三重点でのエントロピーの計算値が0になるようにするために Haar et al. (1984)はSREFの値を7.6180802とおいた。本計算プログラムでは、SREFの値を7.6180720にして基準状態の計算値を0に近い値にした。
T	絶対温度で表示した入力温度
TAU(line 12050)	Wagner and Pruss (1993)の式で用いる温度に関する変数
TAU(line 18350)	$\tau_i$ (HGK式)
TC	純水の臨界温度(= 647.096 K)
TEX	$\exp(-\beta_i \tau_i^2)$ (HGK式)
TIDEAL	$T/100$ (HGK式)
TL	$\log(T/100)$ (HGK式)
TT	温度(°C)の入力値
TTT	絶対温度で表示した入力温度 (HGK式)
TX	$T_i$ (HGK式)
TZ	647.073 (HGK式)
UREF	HGK式では基準状態を三重点での液相にしている。この時に内部エネルギーの計算値が0になるようにしている。Haar et al. (1984)はサブルーチン THERMDTで無次元化した内部エネルギー(UD)の計算値からUREFの値を絶対温度で割った値を引いている。三重点での内部エネルギーの計算値が0になるようにするために Haar et al. (1984)はUREFの値を-4328.455039とおいた。本計算プログラムでは、UREFの値を-4328.454977にして基準状態の計算値を0に近い値にした。
V	$\frac{647.073}{T}$ (HGK式)
V0	純水が気液二相平衡条件下にある時の液相の比体積(cm <sup>3</sup> g <sup>-1</sup> )
V(1),V(2),V(3),V(4), V(5),V(6),V(7),V(8), V(9),V(10)	$b$ や $\bar{B}$ を求めるための係数 (HGK式)
VC	純水の臨界点におけるモル体積の値をJ bar mol <sup>-1</sup> の単位で表した値。単位をcm <sup>3</sup> mol <sup>-1</sup> に取っていない関係でプログラムのline 10450では変数H2Oを変数RHOCで割った値をさらに10で割って変数VCの値としている。
VCON	Haas式における定数(= 3.106 cm <sup>3</sup> g <sup>-1</sup> )
WL1,WL2,WL3,WL4, WL5,WL6	純水が気液二相平衡条件下にある時に液相の密度をWagner-Pruss式で計算するための係数

---

---

WLKCL	液相中の塩化カリウムの濃度(wt%)
WV1,WV2,WV3,WV4, WV5,WV6	純水が気液二相平衡条件下にある時に気相の密度をWagner-Pruss式で計算するための係数
X	純水の密度 (HGK式)
XL	液相中での塩化カリウムのモル分率
XV	気相中での塩化カリウムのモル分率
XX	$1 - y$ (HGK式)
Y	$bp/4$ (HGK式)
YL	液相中でH <sub>2</sub> O 1 モルに溶解している塩化カリウムの物質量 (モル)
YV	気相中でH <sub>2</sub> O 1 モルに溶解している塩化カリウムの物質量 (モル)
YVDEC	$y^{\text{vapor}}$ の初期推定値を求めるための変数
YVMAX	YVの値の最大値
YVMIN	YVの値の最小値
Z (line 19050)	$p_{\text{base}}/\rho RT$ (HGK式)
Z (line 19350)	$p/\rho RT$ (HGK式)
Z0	$\frac{1 + \alpha y + \beta y^2}{(1 - y)^3}$ (HGK式)

---

#### 4. Program list

```
10000 REM Calculation of vapor-liquid equilibrium for aqueous KCl solution
10050 REM Hovey et al. (1990)
10100 DEFDBL A-H, M-Z
10150 DIM HGKG(40), II(40), JJ(40), BP(10), BQ(10)
10200 DIM ATZ(4), ADZ(4), AAT(4), AAD(4)
10250 DIM QR(11), QT(10), QZR(9), QZT(9), V(10), A(8), C(18)
10300 DIM COEF(10, 10), BBC(10), BC(10)
10350 GOSUB *BLOCKDATA
10400 H2O=18.01534:KCL=74.551#:R=8.3144#
10450 RHOC=.322#:VC=H20/(RHOC*10)
10500 INPUT"Input temperature (deg C)";TT
10550 IF TT<300 THEN GOTO 10500
10600 T=TT+273.15#
10650 TTT=T
10700 RT=R*T
10750 INPUT "Input molality of KCl in liquid phase";M
10760 IF M=<0 THEN GOTO 10750
10800 LPRINT USING"Temperature(deg C)=####.###      molality of KCl=+#.#.#####";TT,M
10850 LPRINT
10900 WLKCL=100*KCL*M/(1000+KCL*M)
10950 S=6509050/T/T-29128.5#/T+41.2623#+.0759889*T
11000 PVLS=-308.19+5.01055*TT-.0313178*TT*TT+9.29982D-005*TT*TT*TT
11050 PVLS=PVLS-1.19918D-007*TT*TT*TT*TT+5.46864D-011*TT*TT*TT*TT*TT
11100 IF WLKCL<S THEN GOTO 11350
11150 LPRINT"Supersaturated (or saturated) with KCl"
11200 LPRINT USING"Pressure = +#.###^##### Solubility=+#.#.";PVLS,1000*S/((100-S)*KCL)
11250 LPRINT : LPRINT : LPRINT : LPRINT
11300 GOTO 10500
11350 XL=M/(M+1000/H20)
11400 YL=M*H20/1000
11450 INPUT"Do you input intial estimates for DRL, DRV, and YV? Input Y (or y) or N (n)";A$
11500 IF A$="N" OR A$="n" THEN GOTO 11950
11550 INPUT"DRL";DRL:INPUT"DRV";DRV:INPUT"YV";YV
11600 DELDL=1D-006:DELDV=1D-006
11650 MYULS=0:MYUVS=0:MYULWATER=0:MYUVWATER=0
11700 REM Parameters of Hovey et al.
11750 B10=116989.96#-157.8381#*T+.06641785*T*T-3.039773*10000000/T
11800 B11=-4452.32#+2.11429#*T-1.958284#*100000000000000/(T*T*T*T)
11850 B20=-37956.21#+45.32167#*T-1.831384#*1D+020/(T*T*T*T*T*T)
11900 GOTO 14450
11950 TC=647.096
12000 IF TT=<350 THEN GOTO 12100
12050 TAU=1#-623.15#/TC:GOTO 12150
12100 TAU=1#-T/TC
12150 GOSUB *WPKCL
12200 DRLO=DAQ*(1#-WLKCL/100)/RHOC:DRVO=DWATERV/RHOC
12250 XV=1D-012:YV=XV/(1#+XV)
12300 DRL=0:DRV=0
12350 DELDL=1D-006:DELDV=1D-006
12400 MYULS=0:MYUVS=0:MYULWATER=0:MYUVWATER=0
12450 REM Parameters of Hovey et al.
12500 B10=116989.96#-157.8381#*T+.06641785*T*T-3.039773*10000000/T
12550 B11=-4452.32#+2.11429#*T-1.958284#*100000000000000/(T*T*T*T)
12600 B20=-37956.21#+45.32167#*T-1.831384#*1D+020/(T*T*T*T*T*T)
12650 DRL=DRLO
12700 DRV=DRVO
12750 MYULS=RT*LOG(XL)+VC*((-1#)*B10/DRL+B11*(LOG(DRL)+1#/DRL))-2#*VC*YL*B20/DRL
```

```

12800 MYUVS=RT*LOG(YV)-RT*LOG(1#+YV)+VC*((-1#)*B10/DRV+B11*(LOG(DRV)+1#/DRV))-2#*VC*YV*B20/DRV
12850 DIFMYUS=MYUVS-MYULS
12900 DMYUSDY=RT/YV/(1#+YV)-2#*VC*B20/DRV
12950 IF ABS(DIFMYUS)<1D-010 THEN GOTO 13250
13000 IF ABS(DIFMYUS/DMYUSDY)<1D-021 THEN GOTO 13250
13050 YVDEC=DIFMYUS/DMYUSDY
13100 IF YVDEC<YV THEN GOTO 13200
13150 YVDEC=YVDEC*.5#:GOTO 13100
13200 YV=YV-YVDEC:GOTO 12800
13250 XV=YV/(1#+YV)
13300 X=0:PRES=0:PL=0:PV=0:PLD=0
13350 RHOV=DRV*RHOC
13400 X=RHOV
13450 GOSUB *WATER
13500 PV=PRES
13550 PV=PV+YV*(B10+B11*(DRV-1#))+YV*YV*B20
13600 RHOL=DRL*RHOC
13650 X=RHOL
13700 GOSUB *WATER
13750 PL=PRES
13800 PL=PL+YL*(B10+B11*(DRL-1#))+YL*YL*B20
13850 DIFPLV=PL-PV
13900 RHOLD=(DRL+DELDL)*RHOC
13950 X=RHOLD
14000 GOSUB *WATER
14050 PLD=PRES
14100 PLD=PLD+YL*(B10+B11*(DRL+DELDL)-1#)+YL*YL*B20
14150 DPDDL=(PLD-PL)/DELDL
14200 IF ABS(DIFPLV/DPDDL)<1D-010 THEN GOTO 14450
14250 DRLDEC=DIFPLV/DPDDL
14300 IF DRL>DRLDEC THEN GOTO 14400
14350 DRLDEC=.5#*DRLDEC:GOTO 14300
14400 DRL=DRL-DRLDEC:GOTO 13600
14450 GOSUB *SOLUTION
14500 LPRINT
14550 DRL=BC(1):DRV=BC(2):YV=BC(3)
14600 GOSUB *FUNC
14650 RHOLIQUID=(H20+YL*KCL)*RHOC*DRL/H20
14700 RHOVAPOR=(H20+YV*KCL)*RHOC*DRV/H20
14750 LPRINT USING"Pressure liquid=+#.###^vapor=+#.###^";PL,PV
14800 LPRINT USING"Density liquid=+#.###^vapor=+#.###^";RHOLIQUID,RHOVAPOR
14850 LPRINT:LPRINT:LPRINT:LPRINT
14900 GOTO 10500
14950 *WPKCL
15000 REM Densities of liquid and vapor under vapor-saturation
15050 WL1=1.99274064:WL2=1.09965342:WL3=-.510839303:WL4=-1.75493479#:WL5=-45.5170352:WL6=-674694.45#
15100 WV1=-2.0315024:WV2=-2.6830294:WV3=-5.38626492#:WV4=-17.2991605:WV5=-44.7586581#:WV6=-63.9201063
15150 DL=1#+WL1*EXP((1#/3#)*LOG(TAU))+WL2*EXP((2#/3#)*LOG(TAU))+WL3*EXP((5#/3#)*LOG(TAU))+WL4*EXP((16#/3#)*LOG(TAU))+WL5*EXP((43#/3#)*LOG(TAU))+WL6*EXP((110#/3#)*LOG(TAU))
15200 DL=RHOC*DL
15250 DV=WV1*EXP((2#/6#)*LOG(TAU))+WV2*EXP((4#/6#)*LOG(TAU))+WV3*EXP((8#/6#)*LOG(TAU))+WV4*EXP((18#/6#)*LOG(TAU))+WV5*EXP((37#/6#)*LOG(TAU))+WV6*EXP((71#/6#)*LOG(TAU))
15300 DV=RHOC*EXP(DV):DWATERV=DV
15350 REM Haas equation
15400 BA1=-940.409:BA2=1716.28#:BA3=-774.341#:BA4=-126.905:BA5=116.456#
15450 VCON=3.106
15500 V0=1#/DL
15550 D1=1000*V0+M*BA1+M*BA2*V0+M*BA3*V0*V0+(BA4+BA5*V0)*V0*V0*M*SQR(M)/((VCON-V0)*(VCON-V0))
15600 DAQ=(1000+KCL*M)/D1

```

```
15650 RETURN
15700 *WATER
15750 GIB=0
15800 RTG=GASCON*T
15850 REM BBT
15900 V(1)=1#
15950 FOR I=2 TO 10
16000 V(I)=V(I-1)*TZ/T
16050 NEXT I
16100 B1=BP(1)+BP(2)*LOG(1#/V(2))
16150 B2=BQ(1)
16200 FOR I=3 TO 10
16250 B1=B1+BP(I)*V(I-1)
16300 B2=B2+BQ(I)*V(I-1)
16350 NEXT I
16400 DD=X
16450 D=DD*FD
16500 REM QQTD
16550 QR(1)=0
16600 Q5=0
16650 Q=0
16700 AR=0
16750 E=EXP(-AA*D)
16800 Q10=D*D*E
16850 Q20=1#-E
16900 QR(2)=Q10
16950 V=TZ/T
17000 QT(1)=T/TZ
17050 FOR I=2 TO 10
17100 QR(I+1)=QR(I)*Q20
17150 QT(I)=QT(I-1)*V
17200 NEXT I
17250 FOR I=1 TO INC
17300 K=II(I)+1
17350 L=JJ(I)
17400 QK=CDBL(K)
17450 QZR(K-1)=QR(K+1):QZT(L)=QT(L+1):QZR(K)=QR(K+2):QZT(L+1)=QT(L+2)
17500 QP=HGKG(I)*AA*QZR(K-1)*QZT(L)
17550 Q=Q+QP
17600 AR=AR+HGKG(I)*QZR(K)*QZT(L)/(Q10*QK*RTG)
17650 NEXT I
17700 QP=0
17750 FOR J=37 TO 40
17800 IF HGKG(J)=0 THEN GOTO 18900
17850 K=II(J)
17900 KM=JJ(J)
17910 QK=CDBL(K):QKM=CDBL(KM)
17950 DDZ=ADZ(J-36)
18000 DEL=D/DDZ-1#
18050 IF ABS(DEL)<1D-010 THEN DEL=1D-010
18150 EX1=-AAD(J-36)*(DEL^QK)
18200 DEX=EXP(EX1)*(DEL^QKM)
18250 ATT=AAT(J-36)
18300 TX=ATZ(J-36)
18350 TAU=T/TX-1#
18400 EX2=-ATT*TAU*TAU
18450 TEX=EXP(EX2)
18500 Q10=DEX*TEX
18550 QM=QKM/DEL-QK*AAD(J-36)*(DEL^(QK-1#))
```

```
18600 FCT=QM*D*D*Q10/DDZ
18650 QP=QP+HGKG(J)*FCT
18700 AR=AR+Q10*HGKG(J)/RTG
18750 NEXT J
18800 Q=Q+QP
18850 REM BASEDT
18900 Y=.25#*B1*D
18950 XX=1#-Y
19000 Z0=(1#+G1*Y+G2*Y*Y)/(XX*XX*XX)
19050 Z=Z0+4#*Y*(B2/B1-GF)
19100 AB=-LOG(XX)-(G2-1#)/XX+28.16666667#/(XX*XX)+4#*Y*(B2/B1-GF)+15.16666667#+LOG(D*RTG/.101325)
19200 BASEF=Z
19300 PRES=FP*(RTG*D*BASEF+Q)
19350 Z=BASEF+Q/(RTG*D)
19400 REM THERMDT
19450 TIDEAL=T/100
19500 TL=LOG(TIDEAL)
19550 GI=-C(1)/TIDEAL+C(2)*TL
19600 FOR I=3 TO 18
19650 GI=GI-C(I)*(TIDEAL^CDBL((I-6)))
19700 NEXT I
19750 AI=GI-1#
19800 AD=AB+AR+AI-UREF/T+SREF
19850 GD=AD+Z
19900 GIB=GD*RTG*FH
19950 RETURN
20000 *FUNC
20050 FSUMI=0
20100 GIB=0: PRES=0: PV=0: PL=0: X=0: DIFP=0: DIFMYUS=0: DIFMYUWATER=0
20150 DRL=BBC(1): DRV=BBC(2): YV=BBC(3)
20200 RHOV=DRV*RHOC
20250 X=RHOV
20300 GOSUB *WATER
20350 GIBV=GIB
20400 PV=PRES
20450 PV=PV+YV*(B10+B11*(DRV-1#))+YV*YV*B20
20500 RHOL=DRL*RHOC
20550 X=RHOL
20600 GOSUB *WATER
20650 GIBL=GIB
20700 PL=PRES
20750 PL=PL+YL*(B10+B11*(DRL-1#))+YL*YL*B20
20800 DIFP=PV-PL
20850 RHOVD=(DRV+DEL DV)*RHOC
20900 X=RHOVD
20950 GOSUB *WATER
21000 GIBVD=GIB
21050 PVD=PRES
21100 PVD=PVD+YV*(B10+B11*((DRV+DEL DV)-1#))+YV*YV*B20
21150 DPDDV=(PVD-PV)/DEL DV
21200 DPDY=B10+B11*(DRV-1#)+2#*YV*B20
21250 RHOLD=(DRL+DEL DL)*RHOC
21300 X=RHOLD
21350 GOSUB *WATER
21400 GIBLD=GIB
21450 PLD=PRES
21500 PLD=PLD+YL*(B10+B11*((DRL+DEL DL)-1#))+YL*YL*B20
21550 DPDDL=(PLD-PL)/DEL DL
21600 MYULS=RT*LOG(XL)+VC*((-1#)*B10/DRL+B11*(LOG(DRL)+1#/DRL))-2#*VC*YL*B20/DRL
```

```
21650 MYUVS=RT*LOG(YV)-RT*LOG(1#+YV)+VC*((-1#)*B10/DRV+B11*(LOG(DRV)+1#/DRV))-2#*VC*YV*B20/DRV
21700 DIFMYUS=MYUVS-MYULS
21750 DMYUSDDL=VC*B11/DRL+VC*(B10-B11+2#*YL*B20)/(DRL*DRL)
21800 DMYUSDDV=VC*B11/DRV+VC*(B10-B11+2#*YV*B20)/(DRV*DRV)
21850 DMYUSDY=RT/YV/(1#+YV)-2#*VC*B20/DRV
21900 MYULWATER=GIBL+YL*VC*(B10/DRL+B11*(1#-1/DRL))+2#*VC*YL*YL*B20/DRL+RT*LOG(1#-XL)
21950 MYUVWATER=GIBV+YV*VC*(B10/DRV+B11*(1#-1/DRV))+2#*VC*YV*YV*B20/DRV-RT*LOG(1#+YV)
22000 DIFMYUWATER=MYUVWATER-MYULWATER
22050 DMYUWATERDDL=(GIBLD-GIBL)/DELDL-YL*VC*(B10-B11+2#*YL*B20)/(DRL*DRL)
22100 DMYUWATERDDV=(GIBVD-GIBV)/DELDV-YV*VC*(B10-B11+2#*YV*B20)/(DRV*DRV)
22150 DMYUWATERDY=VC*(B10/DRV+B11*(1#-1/DRV))+4#*YV*VC*B20/DRV-RT/(1#+YV)
22200 FSUMI=(PV/PL-1#)*(PV/PL-1#)+(MYUVS/MYULS-1#)*(MYUVS/MYULS-1#)+(MYUVWATER/MYULWATER-1#)*(MYUVWATER/MYULWATER-1#)
22250 RETURN
22300 *SOLUTION
22350 REM Non-linear equation
22400 REM Main routine
22450 REM Unknowns=KK
22500 EPS=1D-005:IMAX=100:AL=1#:KK=3
22550 DRLMAX=3.5#:DRLMIN=1.5#:DRVMAX=1#:DRVMIN=.02:YVMIN=1D-012
22600 YVMAX=YL-.001
22650 IF TT<350 THEN GOTO 22750
22700 DRLMIN=1#:DRVMAX=1.5#:GOTO 22750
22750 LPRINT USING"Eps=+#.###^"EPS
22800 LPRINT USING"DRLMAX=+#.### DRLMIN=+#.### DRVMAX=+#.### DRVMIN=+#.### YVMAX=+#.### YVMIN=+#.###"
      ;DRLMAX,DRLMIN,DRVMAX,DRVMIN,YVMAX,YVMIN
22850 BC(1)=DRL:BC(2)=DRV:BC(3)=YV
22900 LPRINT"Initial estimate"
22950 FOR I=1 TO KK
23000 LPRINT USING"BC(#)=+#.###^";I,BC(I);
23050 NEXT I
23100 LPRINT
23150 FOR I=1 TO KK:BBC(I)=BC(I):NEXT I
23200 GOSUB *FUNC
23250 LPRINT USING"Initial FSUM=+#.#####";FSUMI
23300 LPRINT
23350 FSUM=FSUMI
23400 ICON=KK:ITER=1
23450 GOSUB *BCSOLV
23500 LPRINT USING"ICON=## FSUM=+#.##### ITERATION=###";ICON,FSUM,ITER
23550 ITER=ITER+1:AL=1#
23600 FOR J=1 TO KK:LPRINT USING"BC(#)=+#.#####";J,BC(J);:NEXT J:LPRINT
23650 IF ITER>IMAX THEN ICON=-1
23700 LPRINT
23750 IF ICON>0 THEN GOTO 23450
23800 IF ICON=0 THEN GOTO 23950
23850 IF ICON=-1 THEN GOTO 23900
23900 LPRINT"Function improvement possible"
23950 LPRINT"Solution"
24000 FOR J=1 TO KK
24050 LPRINT USING"BC(#)=+#.#####";J,BC(J)
24100 NEXT J
24150 RETURN
24200 REM Subroutine BCSOLV
24250 *BCSOLV
24300 KP1=KK+1
24350 FOR I=1 TO KK:FOR J=1 TO KP1:COEF(I,J)=0:NEXT J:NEXT I
24400 FOR I=1 TO KK:BBC(I)=BC(I):NEXT I
24450 GOSUB *FUNC
```



```
24500 COEF(1, KP1)=(-1#)*DIFP:COEF(2, KP1)=(-1#)*DIFMYUS:COEF(3, KP1)=(-1#)*DIFMYUWATER
24550 COEF(1, 1)=(-1#)*DPDDL:COEF(1, 2)=DPDDV:COEF(1, 3)=DPDY
24600 COEF(2, 1)=(-1#)*DMYUSDDL:COEF(2, 2)=DMYUSDDV:COEF(2, 3)=DMYUSDY
24650 COEF(3, 1)=(-1#)*DMYUWATERDDL:COEF(3, 2)=DMYUWATERDDV:COEF(3, 3)=DMYUWATERDY
24700 REM Solve correction equation
24750 FOR ISL1=1 TO KK
24800 ISL2=ISL1+1
24850 FOR ISL3=ISL2 TO KP1
24900 COEF(ISL1, ISL3)=COEF(ISL1, ISL3)/COEF(ISL1, ISL1)
24950 NEXT ISL3
25000 FOR ISL3=1 TO KK
25050 IF ISL1<>ISL3 THEN GOTO 25150
25100 GOTO 25300
25150 FOR ISL4=ISL2 TO KP1
25200 COEF(ISL3, ISL4)=COEF(ISL3, ISL4)-COEF(ISL1, ISL4)*COEF(ISL3, ISL1)
25250 NEXT ISL4
25300 NEXT ISL3
25350 NEXT ISL1
25400 FOR J1=1 TO KK
25450 BBC(J1)=BC(J1)+COEF(J1, KP1)
25500 NEXT J1
25550 ICON=0
25600 IF BBC(1)>DRLMIN THEN GOTO 25700
25650 BBC(1)=DRLMIN:GOTO 25800
25700 IF BBC(1)<DRLMAX THEN GOTO 25800
25750 BBC(1)=DRLMAX
25800 IF BBC(2)>DRVMIN THEN GOTO 25900
25850 BBC(2)=DRVMIN:GOTO 26000
25900 IF BBC(2)<DRVMAX THEN GOTO 26000
25950 BBC(2)=DRVMAX
26000 IF BBC(3)>YVMIN THEN GOTO 26100
26050 BBC(3)=YVMIN:GOTO 26200
26100 IF BBC(3)<YVMAX THEN GOTO 26200
26150 BBC(3)=YVMAX
26200 GOSUB *FUNC
26250 IF FSUMI<FSUM THEN FSUM=FSUMI:GOTO 27400
26300 AL=AL*1#/2#
26350 FOR J1=1 TO KK
26400 BBC(J1)=BC(J1)+AL*COEF(J1, KP1)
26450 NEXT J1
26500 IF BBC(1)>DRLMIN THEN GOTO 26600
26550 BBC(1)=DRLMIN:GOTO 26700
26600 IF BBC(1)<DRLMAX THEN GOTO 26700
26650 BBC(1)=DRLMAX
26700 IF BBC(2)>DRVMIN THEN GOTO 26800
26750 BBC(2)=DRVMIN:GOTO 26900
26800 IF BBC(2)<DRVMAX THEN GOTO 26900
26850 BBC(2)=DRVMAX
26900 IF BBC(3)>YVMIN THEN GOTO 27000
26950 BBC(3)=YVMIN:GOTO 27100
27000 IF BBC(3)<YVMAX THEN GOTO 27100
27050 BBC(3)=YVMAX
27100 GOSUB *FUNC
27150 IF FSUMI<FSUM THEN FSUM=FSUMI:GOTO 27400
27200 AL=AL*.5#
27250 IF AL>1D-008 OR AL<-1D-008 THEN GOTO 26350
27300 IF AL>=-1D-008 AND AL<0 THEN LPRINT "Failed. ":END
27350 AL=-1:GOTO 26350
27400 FOR I=1 TO KK
```

```
27450 IF ABS(COEF(I, KP1)/BBC(I))>=EPS THEN ICON=ICON+1
27500 NEXT I
27550 FOR I=1 TO KK:BC(I)=BBC(I):NEXT I
27600 RETURN
27650 *BLOCKDATA
27700 FOR I=1 TO 4:READ ATZ(I):NEXT I
27750 DATA 640#, 640#, 641.6#, 270#
27800 FOR I=1 TO 4:READ ADZ(I):NEXT I
27850 DATA 0.319#, 0.319#, 0.319#, 1.55#
27900 FOR I=1 TO 4:READ AAT(I):NEXT I
27950 DATA 2.0D+004, 2.0D+004, 4.0D+004, 25.0#
28000 FOR I=1 TO 4:READ AAD(I):NEXT I
28050 DATA 34.0#, 40.0#, 30.0#, 1.050#
28100 GASCON=.461522#:TZ=647.073:AA=1#:INC=36
28150 UREF=-4328.454977#:SREF=7.6180720#
28200 G1=11#:G2=44.333333333333#:GF=3.5#
28250 FOR I=1 TO 10:READ BP(I):NEXT I
28300 DATA 0.7478629#, -0.3540782#, 0.0#, 0.0#, 0.007159876#, 0.0#, -0.003528426#, 0.0#, 0.0#, 0.0#
28350 FOR I=1 TO 10:READ BQ(I):NEXT I
28400 DATA 1.1278334#, 0.0#, -0.5944001#, -5.010996#, 0.0#, 0.63684256#, 0.0#, 0.0#, 0.0#, 0.0#
28450 FOR I=1 TO 40:READ HGKG(I):NEXT I
28500 DATA -5.3062968529023D+002, 2.2744901424408D+003, 7.8779333020687D+002
28550 DATA -6.9830527374994D+001, 1.7863832875422D+004, -3.9514731563338D+004
28600 DATA 3.3803884280753D+004, -1.3855050202703D+004, -2.5637436613260D+005
28650 DATA 4.8212575981415D+005, -3.4183016969660D+005, 1.2223156417448D+005
28700 DATA 1.1797433655832D+006, -2.1734810110373D+006, 1.0829952168620D+006
28750 DATA -2.5441998064049D+005, -3.1377774947767D+006, 5.2911910757704D+006
28800 DATA -1.3802577177877D+006, -2.5109914369001D+005, 4.6561826115608D+006
28850 DATA -7.2752773275387D+006, 4.1774246148294D+005, 1.4016358244614D+006
28900 DATA -3.1555231392127D+006, 4.7929666384584D+006, 4.0912664781209D+005
28950 DATA -1.3626369388386D+006, 6.9625220862664D+005, -1.0834900096447D+006
29000 DATA -2.2722827401688D+005, 3.8365486000660D+005, 6.8833257944332D+003
29050 DATA 2.1757245522644D+004, -2.6627944829770D+003, -7.0730418082074D+004
29100 DATA -0.225#, -1.68#, 0.055#, -93.0#
29150 FOR I=1 TO 40:READ II(I):NEXT I
29200 DATA 0, 0, 0, 0, 1, 1, 1, 1, 2, 2, 2, 2, 3, 3, 3, 3, 3, 3, 4, 4, 4, 4, 4, 5, 5, 5, 5, 6, 6, 6, 6, 8, 8, 8, 8, 2, 2, 0, 4, 2, 2, 2, 4
29250 FOR I=1 TO 40:READ JJ(I):NEXT I
29300 DATA 2, 3, 5, 7, 2, 3, 5, 7, 2, 3, 5, 7, 2, 3, 5, 7, 2, 3, 5, 7, 2, 3, 5, 7, 2, 3, 5, 7, 2, 3, 5, 7, 1, 4, 4, 4, 0, 2, 0, 0
29350 FOR I=1 TO 18:READ C(I):NEXT I
29400 DATA 1.9730271018D+001, 2.09662681977D+001, -4.83429455355D-001, 6.05743189245D+000, 2.256023885D+001, -9.87
532442D+000, -4.3135538513D+000, 4.58155781D-001
29450 DATA -4.7754901883D-002, 4.1238460633D-003, -2.7929052852D-004
29500 DATA 1.4481695261D-005, -5.6473658748D-007, 1.6200446D-008, -3.303822796D-010, 4.51916067368D-012, -3.707341
22708D-014, 1.37546068238D-016
29550 FD=1#:FP=10:FH=18.01534
29600 RETURN
```

## 文献

澁江靖弘 (2012) 300°Cから410°Cにおける塩化カリウム—水系の気液平衡. 兵庫教育大学研究紀要, **40**, 79–91.